

# *Occultations de planètes par la Lune*

## *Ephémérides astronomiques*

*Serveur de l'IMCCE*

### **Mouvement de la Lune**

La position du plan de l'orbite de la Lune inclinée de  $5^{\circ}15'$  sur l'écliptique, et les variations rapides des éléments de celle-ci (nœuds, périégée...) font qu'au cours des révolutions successives, la trajectoire apparente de la Lune sur le ciel se modifie sans cesse.

En quelques années, la Lune a balayé une surface non négligeable du ciel, cachant successivement pendant quelques minutes à plus d'une heure les objets situés sur sa trajectoire.

Constamment la Lune occulte des étoiles et plus rarement des planètes.

Le phénomène de disparition et de réapparition pour les objets brillants est facile à observer surtout s'il se produit du côté de l'ombre.

La Lune étant un objet proche, le phénomène observable par effet de parallaxe est variable en position et temps avec la position de l'observateur.

Grâce aux éphémérides précises topocentriques de l'IMCCE, nous allons pouvoir tracer le mouvement apparent de l'astre par rapport à la Lune, et prédire avec une bonne précision les positions et les instants de disparition (Immersion) et de réapparition (Émersion).

Pour les étoiles le phénomène est extrêmement bref. Pour les planètes il dure plusieurs secondes. Il ne sera pas tenu compte du diamètre apparent de la planète qui allonge le temps d'immersion et d'émersion, mais cela peut être envisagé.

Les occultations de planètes par la Lune peuvent être étendues aux occultations d'étoiles. Le problème est plus simple car durant l'occultation, l'étoile n'a pas de mouvement sur le ciel. Mais il faut aller chercher la position de l'étoile hors IMCCE et être bien compatible en coordonnées.

### **Ephémérides**

L'événement étudié sera l'occultation du 2 mars 2007 à 3h TU.

Nous nous plaçons dans le repère topocentrique en coordonnées équatoriales ou locales (ou horaires).

Les éphémérides donnent à tout instant choisi :

- la direction de la planète  $\alpha_p$  et  $\delta_p$  (alpha et delta) ou  $a_p$  et  $h_p$  (azimut et hauteur)
- la direction de la Lune  $\alpha_L$  et  $\delta_L$  ou  $a_L$  et  $h_L$

Nous connaissons :

- le diamètre de la Lune  $R_L$

### **Scénario**


Pour suivre l'occultation, nous allons calculer avec les éphémérides, pas à pas :

- $\theta$  l'angle entre la direction du centre de la Lune et la direction de la planète.
- $\rho_L$  le rayon angulaire de la Lune

Et tracerons les graphes des variations de ces grandeurs en fonction du temps

A partir des données, nous chercherons

- les instants où la planète touche le bord de la Lune
- par le calcul le moment précis de contact
- l'angle de position de la planète relativement à la direction polaire ou zénithale lors des contacts.



## Plan de travail

- 1) Téléchargement depuis le site de l'IMCCE des éphémérides de la Lune et de la planète.
  - Coordonnées topocentriques du lieu d'observation
  - Coordonnées équatoriales ou horaires ou locales
  - Coordonnées Apparentes (équateur vrai ; équinoxe de la date)
  - De 2 heures avant à 2 heures après par rapport au milieu de l'occultation, pas de 1 minute ou 30 secondes
- 2) Transformation des données :
  - dates, heures
  - coordonnées topocentriques
  - distance pour la Lune
- 3) Calculs :
  - Différences de coordonnées et distance angulaire
  - Rayon angulaire de la Lune
  - Recherche des contacts et ajustement pour les calcul des instants
  - Calcul des angles au pôle ou au zénith
- 4) Tracé des graphiques

## Fichier tableur et feuilles de calcul

Le stockage et la manipulation des données se fait au moyen d'un tableur, en l'occurrence Excel. Tout est transposable dans d'autres programmes.


Pour l'uniformisation des différents fichiers d'éphémérides, chacun comprend

- une feuille d'éphémérides pour chaque objet ([Lune] et [Planète])
- une feuille de calculs et de graphiques ([Occultation]).

## Données IMCCE (<http://www.imcce.fr>)

Les pages de résultats des éphémérides de l'IMCCE comportent une partie de mise en page encadrant les résultats. Les données utiles commencent à la première ligne de « # » et finissent à la dernière ligne d'éphémérides.

Partie à sélectionner  
à copier, coller  
dans les feuilles Excel



IMCCE INSTITUT DE MÉCANIQUE CÉLESTE ET DE CALCUL DES ÉPHÉMÉRIDES

ÉPHÉMÉRIDES → RESULTATS

• Senseur d'éphémérides de l'IMCCE • Signification des paramètres calculés  
• Fichier résultats au format asii

ÉPHÉMÉRIDES DES CORPS DU SYSTÈME SOLAIRE

Satellite 10 Lune  
Théorie planétaire DE405/LE405  
Coordonnées Astronomiques J2000  
Centre du repère : géocentre  
Coordonnées écliptiques (lambda, beta)

Date UTC	h	m	s	o	Long.	o	Lat.	o	Distance	V.Mag	Phase	Elong.
					°	'	°	'	Rt.		o	°
3 3 2007 20 0 0.00	161	14	23.0806	+00	26	31.2304	63.013868989	-12.52	1.58	178.41		
3 3 2007 20 0 30.00	161	14	36.1491	+00	26	29.8401	63.013985541	-12.52	1.58	178.42		
3 3 2007 20 1 0.00	161	14	53.2175	+00	26	28.4497	63.014102086	-12.52	1.57	178.42		
3 3 2007 20 1 30.00	161	15	8.2858	+00	26	27.0594	63.014218622	-12.52	1.57	178.43		
3 3 2007 20 2 0.00	161	15	23.3541	+00	26	25.6691	63.014335149	-12.52	1.57	178.43		
3 3 2007 20 2 30.00	161	15	38.4223	+00	26	24.2787	63.014451667	-12.52	1.56	178.43		
3 3 2007 20 3 0.00	161	15	53.4905	+00	26	22.8884	63.014568177	-12.52	1.56	178.44		
3 3 2007 20 3 30.00	161	16	8.5586	+00	26	21.4980	63.014684678	-12.53	1.56	178.44		
3 3 2007 20 4 0.00	161	16	23.6266	+00	26	20.1077	63.014801170	-12.53	1.55	178.44		
3 3 2007 20 4 30.00	161	16	38.6946	+00	26	18.7173	63.014917653	-12.53	1.55	178.45		
4 3 2007 4 15 0.00	165	22	35.7057	+00	3	34.2476	63.124885992	-12.47	2.26	182.27		
4 3 2007 4 15 30.00	165	22	48.7194	+00	3	32.8573	63.124983606	-12.47	2.27	182.27		
4 3 2007 4 16 0.00	165	23	3.7337	+00	3	31.4670	63.125101220	-12.47	2.27	182.28		
4 3 2007 4 16 30.00	165	23	18.7480	+00	3	30.0767	63.125208825	-12.47	2.27	182.28		
4 3 2007 4 17 0.00	165	23	33.7622	+00	3	28.6864	63.125316421	-12.47	2.28	182.28		
4 3 2007 4 17 30.00	165	23	48.7763	+00	3	27.2962	63.125424007	-12.47	2.29	182.29		
4 3 2007 4 18 0.00	165	24	3.7904	+00	3	25.9059	63.125531584	-12.47	2.29	182.29		
4 3 2007 4 18 30.00	165	24	18.8045	+00	3	24.5157	63.125639152	-12.47	2.29	182.30		
4 3 2007 4 19 0.00	165	24	33.8185	+00	3	23.1254	63.125746711	-12.47	2.29	182.30		
4 3 2007 4 19 30.00	165	24	48.8324	+00	3	21.7351	63.125854260	-12.47	2.30	182.30		

Remplir et soumettre le formulaire pour calculer une nouvelle éphéméride:

Charger un fichier de dates...

ou

Année Mois Jour Heure Minutes Secondes  
 2007 03 03 20 Date courante Effacer

• Nombre de dates: 1000 (+ 5000)

• Pas de calcul: 30 secondes

• Format des dates ? Date gregorienne

• Format des sorties ? asii

Attention: Ce sensur d'éphémérides a été mis en place avec l'aide du Ministère de l'Éducation Nationale, du CNRS et du CNES. Si vous devez utiliser ces données en vue d'une diffusion publique, veuillez nous consulter

retour haut de page |  
 imprimer la page |  
 paramétrer le paramètre

Pour simplifier la manipulation des les données, il n'y a pas de liens entre les feuilles d'éphémérides de chaque corps, et les données de la feuille de calculs. Libre à l'utilisateur de le faire.

En conséquence, les éphémérides de la feuille « Occultation » sont à prendre dans les feuilles des corps et à reporter dans la feuille de « Occultation » avec « copié-collé spécial valeurs ».

## Calcul d'éphémérides à l'IMCCE

Pour la Lune et la planète :

- ① Un corps du Système solaire    Lune, Soleil
- ② La théorie planétaire            garder celle par défaut
- ③ Centre du repère                topocentrique
- ④ Plan de référence                équateur
- ⑤ Type de coordonnées            sphériques
- ⑥ Type d'éphémérides            Astrométrique J2000
- ⑦ L'échelle de temps                UTC
- ⑧ L'époque du calcul  
première date                    2007 03 02 02 00 00  
pas échantillonnage            30s  
nombre de points                500
- ⑨ Lancer le calcul

Entrer à la demande, le département, la ville choisie ou les coordonnées précise d'un lieu.

## Sauvegarde des résultats

Ouvrir Excel et créer 3 feuilles que l'on renommerà :

*Occultation*  
*Lune*  
*Planète*

Pour la Lune et la planète, faire un copié-collé des données obtenues sur le site de l'IMCCE dans la colonne A de la feuille correspondante Excel.

Pour la partie à sélectionner voir page précédente.

**Note** : si dans Excel, le *caractère décimal* des feuilles de calcul n'est pas le point décimal vous pouvez soit

- dans un Excel mettre le point comme caractère décimal,
- dans les éphémérides de la colonne A, changer les points en virgules.

Fichiers Excel de travail :

Fichier résultat complet    occult\_20070302.xls  
Fichier structure seule        occult\_20070302-0.xls

à prendre sur le site

[http://www-obs.univ-lyon1.fr/fc/ateliers\\_2006-07/ateliers.htm](http://www-obs.univ-lyon1.fr/fc/ateliers_2006-07/ateliers.htm)

Page web à remplir pour le calcul

## Conversion des données en valeurs numériques

Les formules de calcul s'appliquent aux cellules de la ligne 16. Il suffit ensuite d'appliquer par [sélectionner glisser] pour l'appliquer aux autres lignes de données, ce qui incrémente le rang des cellules dans les formules.

### > Dates et heures ([colonne B et C])

Les dates étant communes aux deux objets, il suffit de les extraire une fois pour la Lune ([colonnes B et C]) et les reporter dans les colonnes A et B de la "feuille Eclipse" par un collé spécial valeurs.

Dates : formule [cellule B16]

=DATE(CTXT(STXT(A17;9;4));CTXT(STXT(A17;4;3));CTXT(STXT(A17;1;3)))

Heures : formule [cellule C16]

=(CNUM(STXT(A17;14;2))+CNUM(STXT(A17;17;2))/60+CNUM(STXT(A17;20;5))/3600)/24

Ne pas oublier de mettre le **format Date** dans la [colonne B] et **format Heure** dans la [colonne C]. Le formatage (hh:mm) suffit.

### > coordonnées ([feuille Lune et Saturne colonnes D et E])

- Ascensions droites

=CNUM(STXT(A17;27;3))+CNUM(STXT(A17;31;2))/60+CNUM(STXT(A17;34;7))/3600

- Déclinaisons

=(CNUM(STXT(A17;44;2))+CNUM(STXT(A17;47;2))/60+CNUM(STXT(A17;50;7))/3600)\*  
(SI(STXT(A17;43;1)="-";-1;1))

ou

- Azimuts

=CNUM(STXT(A17;27;3))+CNUM(STXT(A17;31;2))/60+CNUM(STXT(A17;34;7))/3600

- Hauteurs

=(CNUM(STXT(A17;42;2))+CNUM(STXT(A17;45;2))/60+CNUM(STXT(A17;48;7))/3600)\*  
(SI(STXT(A17;41;1)="-";-1;1))

## Données générales

Mettre dans les cellules de la [feuille Lune] :

- [cellule B6] : le rayon de la Lune 1737.4 km
- [Cellule B8] : l'unité astronomique en kilomètre 149597870 km

Dans les formules, pour leur utilisation, il faudra l'écrire avec \$ (\$C\$L) pour éviter les décalages.

## Disposition feuille Lune et Saturne – Données tirées des éphémérides

Colonne	Feuille Saturne	Feuille Lune
A	Données IMCCE	Données IMCCE
B	Dates	Dates
C	heures	heures
D	Coordonnées $\alpha$ ou $a$	Coordonnées $\alpha$ ou $a$
E	Coordonnées $\delta$ ou $h$	Coordonnées $\delta$ ou $h$
F		Distance (unités astronomiques)

### [Feuille Lune]

A													B	C	D	E	F	G			
EPHEMERIDES DES CORPS DU SYSTEME SOLAIRE																					
Satellite 10 Lune													R Lune (km)								
Theorie planetaire DE405/LE405													1737.40								
Coordonnees Apparentes (equateur vrai : equinoxe de la date)													1 UA (km)								
Centre du repere : topocentre : ST-GENIS-LAVAL													149597870								
0 h 19 m 8.000 s E : 45 d 42 ' 0.000 " N : 0.00 m																					
Coordonnees equatoriales (R.A, Dec.)																					
Date UTC													Date	Heure	AR	DEC	Dist. (ua.)	Rayon ang.			
h	m	s	h	m	s	o	'	"	ua.	V.Mag	Phase	Elong.	muRAcosDE	muDE							
													" / s	" / s							
2	3	2007	0	0	0.00	9	27	49.56982	+16 58 22.8784	0.002620225	-11.80	22.14	157.91	0.323E+00	-0.223E+00	02/03/2007	0:00:00	9.463769	16.97302	0.0026202	0.26395432
2	3	2007	0	1	0.00	9	27	50.91954	+16 58 9.5129	0.002620286	-11.80	22.14	157.91	0.323E+00	-0.223E+00	02/03/2007	0:01:00	9.464144	16.96931	0.0026203	0.26394841
2	3	2007	0	2	0.00	9	27	52.27034	+16 57 56.1354	0.002620347	-11.80	22.13	157.92	0.323E+00	-0.223E+00	02/03/2007	0:02:00	9.464520	16.96559	0.0026203	0.2639425

Dans la [feuille Lune colonne G] nous ferons un premier calcul sur tous les rangs, celui du rayon angulaire de la Lune  $\rho_{\text{Lune}}$ .

Formule : =DEGRES(ATAN(\$B\$6/(\$B\$8\*F17)))

### Disposition feuille Eclipse – Données tirées des éphémérides

Colonne	Contenu	Provenance par Copié-collé spécial valeurs
A	Dates	[Feuille Lune col. B]
B	heures	[Feuille Lune col. C]
D	Coordonnées $\alpha$ ou $a$ Saturne	[Feuille Saturne col. D]
E	Coordonnées $\delta$ ou $h$ Saturne	[Feuille Saturne col. E]
F	Coordonnées $\alpha$ ou $a$ Lune	[Feuille Lune col. D]
G	Coordonnées $\delta$ ou $h$ Lune	[Feuille Lune col. E]

A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
le 02 mars 2007														
					Immersion Emersion									
					Heure	0.1250829	0.1284391							
					Angle pôle	206.6825042	215.5202592							
Tracé cercle lunaire														
												moyenne R Lune 15.13454		
Date		Heure	Heure	Saturne	Saturne	Lune	Lune	Position Saturne / Lune		Rayon ang.		$\beta$	x	y
		hh.mm	continue	AR	DEC	AR	DEC	$\Delta\alpha \cos \delta$	$\Delta\delta$	$\theta$	$\rho_{\text{Lune}}$	(°)	(°)	(°)
02/03/2007		0.00	0.00	9.5316356	16.0314654	9.4637694	16.9730216	58.7042	-56.4934	81.4720	15.2373	0	15.13454	0.00000
02/03/2007		0.01	0.01	9.5316322	16.0314819	9.4641443	16.9693091	58.3770	-56.2696	81.0811	15.2369	1	15.13223	0.26413

### Disposition feuille Eclipse – Données calculées à partir des éphémérides

Col.	Contenu	Calcul
C	heures continues	calculées pour le problème de changement de jour
H	$\Delta\alpha \cos \delta$ ou $\Delta a \cos h$	en minutes d'arc
I	$\Delta\delta$ ou $\Delta h$	en minutes d'arc
J	$\theta$ distance angulaire, axe du cône Lune	en minutes d'arc
K	diamètre angulaire de la Lune $\rho_{\text{Lune}}$	en minutes d'arc [Feuille Lune col. G]
M	angle de position sur le cercle lunaire	Tracé du cercle lunaire en minutes d'arc
N	abscisses des points du cercle lunaire	
O	ordonnées des points du cercle lunaire	

A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
le 02 mars 2007														
					Immersion Emersion									
					Heure	0.1250829	0.1284391							
					Angle pôle	206.6825042	215.5202592							
Tracé cercle lunaire														
												moyenne R Lune 15.13454		
Date		Heure	Heure	Saturne	Saturne	Lune	Lune	Position Saturne / Lune		Rayon ang.		$\beta$	x	y
		hh.mm	continue	AR	DEC	AR	DEC	$\Delta\alpha \cos \delta$	$\Delta\delta$	$\theta$	$\rho_{\text{Lune}}$	(°)	(°)	(°)
02/03/2007		0.00	0.00	9.5316356	16.0314654	9.4637694	16.9730216	58.7042	-56.4934	81.4720	15.2373	0	15.13454	0.00000
02/03/2007		0.01	0.01	9.5316322	16.0314819	9.4641443	16.9693091	58.3770	-56.2696	81.0811	15.2369	1	15.13223	0.26413



Dans l'encadré [cellules E3 à G5], seront les résultats : les calculs des instants des contacts et les angles aux pôles (ou au zénith).

### Heures continues

Une occultation suivant la distance au centre lunaire peut durer plus d'une heure à quelques instants. Habituellement, ceci se passe dans une même journée. Mais il arrive que l'occultation commence un jour et finisse le lendemain.

Dans ce cas, il y a un saut de 24 à 0 heures dans les valeurs de l'heure, ce qui est très gênant pour tracer un graphe avec le temps comme abscisses.

Sous Excel les heures sont codées en jours décimaux (0h ⇒ 0.0, 1h ⇒ 0.041666, 2h ⇒ 0.08333, 3h ⇒ 0.25 etc, mais apparaissent sous la forme conventionnelle hh:mm:ss par le formatage.

Pour éviter le problème au passage à 0h, on ajoute 1 (24h).

Dans la [feuille Occultation] on crée une [colonne C] heure continue :

S'il y a passage à 24 heures, l'heure continue de croître : 24h, 25h, 26h... mais s'affichera 0h, 1h, 2h... avec le format heure d'Excel.

Ceci se traduit par la formule de la [cellule C17] : =(A17-\$A\$17+B17), à reporter dans toutes les cellules de la [colonne C]. Cette formule permet de travailler en continu sur plusieurs jours.

### Cercle lunaire

Pour mieux visualiser, l'occultation, il y a lieu de tracer sur le graphique du déplacement relatif de Saturne par rapport à la Lune, le cercle lunaire.

Son diamètre angulaire varie, mais entre le début et la fin de l'occultation, on peut le considérer comme fixe et prendre sa valeur moyenne.

Repérer les valeurs du rayon angulaire de la Lune [col. K] quand Saturne est occulté ( $\theta < \rho_{Lune}$ ) et calculer la moyenne dans la [cellule N12] =MOYENNE(K198:K201)).

Sur deux colonne, on va calculer pour des angles  $\beta$  allant de 0 à 360°, les valeurs de

$$x = \rho_{Lune} \cos \beta \quad \text{et} \quad y = \rho_{Lune} \sin \beta$$

que l'on tracera sur le graphique.

Construction des points du cercle

[col. M] angles en degré 0, 1, 2, ... . 360

[col. N]  $x = \rho_{Lune} \cos \beta = \$N\$12 * \text{COS}(\text{RADIANS}(M17))$

[col. O]  $x = \rho_{Lun} \sin \beta = \$N\$12 * \text{SIN}(\text{RADIANS}(M17))$

		Colonnes		
		M	N	O
Rangs		Tracé cercle lunaire		
10				
11		moyenne		
12		R Lune		
13		15.13454		
14		$\beta$	x	y
15		(°)	(°)	(°)
16				
17	0	15.13454	0.00000	
18	1	15.13223	0.26413	
19	2	15.12532	0.52819	
20	3	15.11380	0.79208	
...	4	15.09767	1.05573	
	5	15.07695	1.31906	
	6	15.05163	1.58199	
	7	15.02173	1.84444	
	8	14.98725	2.10632	
	9	14.94821	2.36756	
	10	14.90461	2.62808	
	11	14.85647	2.88781	
	12	14.80381	3.14665	

### Que peut-on calculer et tracer

De la Terre on va donc voir la Lune s'approcher de Saturne, la cacher ou l'occulter (Immersion). Quelques temps après, Saturne va réapparaître (Emersion).

Le bord de la Lune n'étant pas un cercle parfait, le relief lunaire peut provoquer suivant la trajectoire, des immersions et émerisions successives, surtout quand l'occultation est rasante.

Graphiquement, il va falloir suivre l'évolution dans les temps de la distance angulaire de Saturne au centre de la Lune. Quand cette distance égale le rayon lunaire, nous avons les instants d'immersion et d'émerision.

Les éphémérides n'étant pas continues, il va falloir rechercher par ajustement ou interpolation, les instants précis des contacts avec le bord théorique de la Lune.

Pour mieux prévoir les observations, nous calculerons les positions d'immersion et d'émerision soit par rapport à la direction du Nord (coordonnées équatoriales) soit par rapport au Zénith (coordonnées locales).

### Calcul de la distance Centre Lune - Saturne

Calcul des différences des deux coordonnées dans les [colonnes H et I]:

[colonnes H] (alpha)  $\Delta\alpha \cos \delta = (D17-F17)*15*\text{COS}(\text{RADIANS}(E17))*60$

(azimut)  $\Delta\alpha \cosh = (D17-F17)*\text{COS}(\text{RADIANS}(E17))*60$


[colonnes I]  $\Delta\delta \text{ ou } \Delta h = (E17-G17)*60$

L'angle  $\theta$  se calcule de deux façons :

a) Façon approchée :

$$\theta = \sqrt{(\Delta\alpha \cos \delta)^2 + \Delta\delta^2} \quad \text{ou} \quad \theta = \sqrt{(\Delta a \cos h)^2 + \Delta h^2}$$

car les angle  $\Delta\alpha$  et  $\Delta\delta$  sont petits.



Le calcul de  $\theta$  se fait dans la [colonne J] que l'on exprimera en minutes d'arc.

Formule de calcul : = RACINE(H17^2+I17^2)

b) Formule rigoureuse :

$$\cos \theta = \sin \delta_1 \cdot \sin \delta_2 + \cos \delta_1 \cdot \cos \delta_2 \cdot \cos(\alpha_1 - \alpha_2)$$

$$\cos \theta = \sin h_1 \cdot \sin h_2 + \cos h_1 \cdot \cos h_2 \cdot \cos(a_1 - a_2)$$

## Tracés

Graphique 1 : évolution de  $\Delta\alpha$  (ou  $\Delta a$  ,  $\Delta\delta$  (ou  $\Delta h$ ),  $\theta$ , et  $\rho_{\text{Lune}}$  en fonction du temps

$\Delta\alpha \cos \delta$ ou $\Delta a \cos h$	[colonne C] versus [colonne H]	de 17 à 516
$\Delta\delta$ ou $\Delta h$	[colonne C] versus [colonne I]	de 17 à 516
$\theta$	[colonne C] versus [colonne J]	de 17 à 516
$\rho_{\text{Lune}}$	[colonne C] versus [colonne K]	de 17 à 516

Graphique 2a : déplacement relatif de Saturne par rapport au cercle lunaire : vue d'ensemble

Déplacement :  $\Delta\alpha \cos \delta$  versus  $\Delta\delta$  [colonne H] versus [colonne I] (17 à 516)

Cercle lunaire : [colonne N] versus [colonne O] (17 à 377)

Graphique 2b : déplacement relatif de Saturne par rapport au cercle lunaire : vue détaillée

Idem que graphe 2a, avec des limites d'axes sur la partie occultation.

**Remarque importante** : en coordonnées équatoriales, pour les graphiques 2a et 2b, il faut inverser le sens des abscisses  $\Delta\alpha$ , car les ascensions droites croissent de droites à gauche.

## Calcul des instants des contacts

On recherchera donc le moment de contact  $t_c$  par la recherche de l'intersection de deux droites. L'instant cherché  $t_c$  sera l'abscisse de l'intersection.

- Equations des deux droites

$$\frac{y - y_a}{t - t_a} = \frac{y_b - y_a}{t_b - t_a} \quad \text{et} \quad \frac{z - z_a}{t - t_a} = \frac{z_b - z_a}{t_b - t_a}$$

- Abscisse de l'intersection

$$t = (t_b - t_a) \frac{z_a - y_a}{y_b - y_a - z_b + z_a} + t_a$$

- Ordonnée à l'intersection

$$y_c = z_c = y_a + \frac{t_c - t_a}{t_b - t_a} \cdot (y_b - y_a) = z_a + \frac{t_c - t_a}{t_b - t_a} \cdot (z_b - z_a)$$

Ici, les deux variables x et y sont  $\theta$  et  $\rho_{\text{Lune}}$ .


Nota : on peut créer une colonne différence  $\theta - \rho_{\text{Lune}}$  et rechercher l'instant de la valeur zéro de cette variable.

L'équation devient :

$$\frac{y - y_a}{t_c - t_a} = \frac{y_b - y_a}{t_b - t_a}, \quad \text{avec} \quad y = 0$$

$$t_c = t_a - y_a \frac{t_b - t_a}{y_b - y_a}$$

Formule plus facile à transcrire sous excel.



## Cellules résultats

Ayant repéré les rangs où ont lieu les contacts :

Immersion : entre les rangs 197 et 198

Emersion : entre les rangs 201 et 202

on va écrire les formules du calcul des temps dans les cellules F4 et G4, application directe des formules précédentes.

	A	B	C	D	E	F	G
1	Occultation de Saturne par la Lune						
2	le 02 mars 2007						
3						Immersion	Emersion
4						Heure	3:00:07 3:04:57
5						Angle pôle	206.5 213.8

Les formules des cellules résultats seront :

Immersion [cellule F4]  $= (C198 - C197) * (K197 - J197) / (J198 - J197 - K198 + K197) + C197$

Emersion [cellule G4]  $= (C202 - C201) * (K201 - J201) / (J202 - J201 - K202 + K201) + C201$

## Angles au pôle ou angle au zénith

Les angles au pôle ou au zénith permettent de mieux situer les positions où auront lieu l'immersion et l'émersion.

Soient  $t_a$  et  $t_b$ , les instants dans les éphémérides qui encadrent le temps de contact  $t_c$ ,  $(x_a, y_a)$  et  $(x_b, y_b)$  les positions relatives ( $\Delta\alpha$  ou  $\Delta a$  et  $\Delta\delta$  ou  $\Delta h$ ) de la planète par rapport à la Lune.

On calcule  $x_c$  et  $y_c$ , la position au contact.



$$x_c = x_a + \frac{x_b - x_a}{t_b - t_a} (t_c - t_a) \text{ et } y_c = y_a + \frac{y_b - y_a}{t_b - t_a} (t_c - t_a)$$

L'angle  $\beta$  vaut :  $\beta = \tan^{-1} \frac{y_c}{x_c}$

On en déduit l'angle au zénith :  $A_z = \beta - 90$

ou l'angle au pôle :  $A_p = 360 - (\beta - 90)$

pour tenir compte du sens inverse des ascensions droites.



Formules pour alpha et l'immersion :

$x_{im} : = (H198 - H197) / (C198 - C197) * (P197 - C197) + H197$

$y_{im} : = (I198 - I197) / (C198 - C197) * (P197 - C197) + I197$

$\beta : = \text{DEGRES}(\text{ATAN2}(R197; S197))$

$P : = 360 - \text{MOD}(\text{DEGRES}(\text{ATAN2}(R197; S197)) + 360 - 90; 360)$

Pour être sûr d'avoir des angles positifs, il faut dans les formules, ajouter 360 et en faire le modulo.

Connaissant les points de contacts, on peut dans les graphiques 2a et 2b, tracer les positions des points de contacts et matérialiser les angles par des segments de droites (voir fichier excel).

## Disposition et programmation

Dans excel faire un encadré spécial de cellules pour mettre les calculs à la hauteur des contacts :

	H	I	J	K	L	M	N	O	P	Q	R	S	T	U
195	-5.9261	-14.0525	15.2510	15.1378		178	-15.12532	0.52819						
196	-6.3365	-13.8030	15.1880	15.1371		179	-15.13223	0.26413			$x_{im}$	$y_{im}$	angle $\beta$	angle P
197	-6.7474	-13.5535	15.1401	15.1363		180	-15.13454	0.00000	3:00:07		-6.79652	-13.52366	-116.682504	206.68
198	-7.1590	-13.3038	15.1077	15.1356		181	-15.13223	-0.26413						
199	-7.5712	-13.0541	15.0908	15.1349		182	-15.12532	-0.52819						
200	-7.9840	-12.8043	15.0895	15.1342		183	-15.11380	-0.79208			$x_{em}$	$y_{em}$	angle $\beta$	angle P
201	-8.3975	-12.5544	15.1040	15.1334		184	-15.09767	-1.05573	3:04:57		-8.79179	-12.31642	-125.520259	215.52
202	-8.8116	-12.3045	15.1342	15.1327		185	-15.07695	-1.31906						



On y inscrira les formules :

Immersion		
Calcul	Cellule	Formule
Temps	[P197]	=(B198-B197)*(K197-J197)/(J198-J197-K198+K197)+B197
X	[R197]	=(H198-H197)/(C198-C197)*(P197-C197)+H197
Y	[S197]	=(I198-I197)/(C198-C197)*(P197-C197)+I197
$\beta$	[T197]	=DEGRES(ATAN2(R197;S197))
angle P	[U197]	=360-MOD(DEGRES(ATAN2(R197;S197))+360-90;360) ou =360-MOD(T197)+360-90;360)

Idem pour l'émersion avec les rangs 200 et 201.

Les formules des [cellules P197] et [P201] sont les mêmes que les cellules du tableau de résultats [F5] et [G5].

## Occultation d'étoiles par la Lune

La démarche suivie peut s'appliquer aux occultations d'étoiles.

Le problème est plus simple, l'étoile n'ayant pas de déplacement.

- 1) Trouver dans un catalogue (Internet ou autre) les coordonnées précises de l'étoile (2000).
- 2) Les transformer en coordonnées écliptiques (ou travailler en coordonnées équatoriales).
- 3) Télécharger les éphémérides de la Lune (en système astrométrique 2000).
- 4) Remplacer les coordonnées de la planète par les coordonnées de l'étoile dans toutes les lignes.

ou

faire disparaître les coordonnées de la planète, mais adapter les formules de calcul où les coordonnées de la planète apparaissaient.

### Occultation d'étoiles par la Lune : à quoi ça sert ?

Le suivi de la mesure du flux d'une étoile lors d'une occultation revient à observer les franges de diffraction d'une source quasi ponctuelle par le bord d'un écran. Au lieu d'avoir une image spatiale de la répartition des franges, il est observé une coupe temporelle du phénomène.

L'espacement des franges est proportionnel à la longueur d'onde observée. C'est pourquoi, les premières observations furent faites en ondes radiométriques, puis en infrarouge puis en visible à cause des conditions techniques.

En infrarouge et visible, le phénomène se passe sur quelques centièmes de secondes. Il faut donc un échantillonnage du signal très rapide, de l'ordre de quelques millièmes de secondes.

L'intensité des franges et leur espacement sont directement liés à l'étendue de la source.

On peut ainsi mesurer des diamètres angulaires d'étoiles de l'ordre de quelques milliarcsecondes.

Dans les illustrations ci-dessous on voit :

- 1) deux exemples de signaux d'occultation et l'influence de la taille angulaire de l'objet.
- 2) de l'analyse de l'espacement et de l'intensité des franges de diffraction, la taille de l'objet avec présence d'enveloppe étendue dissymétrique.

Illustrations tirée de *The Messenger* (revue de l'ESO) n° 126 Dec., 2006 p24-26




Figure 3: The light curves (top, blue) and best fit models (black) for 2MASS17474895-2835083 and 2MASS17582187-2814522. The bottom panels show the fit residuals (red), enlarged by a factor of four for clarity. The star on the left is unresolved, with an upper limit of the diameter of 0.65 mas. The other one is resolved, with an estimated diameter of  $3.67 \pm 0.56$  mas. The difference between the two cases is in the number and amplitude of the *diffraction fringes*.




Figure 4: Examples of brightness profiles of stars surrounded by what are presumably compact dust shells. Left, the carbon star C2490 (2MASS17531817-2849492). Right, 2MASS17553507-2841150. In both cases, the inner rims of an optically thin shell are visible, indicating a non-symmetric structure and a characteristic size of about 20 mas.